

マルチガスセンサシステムによる 室内空気汚染ガス種の認知と濃度測定

正員 大藪 多可志 (富山国際大学)
正員 広林 茂樹 (金沢大学)
非会員 木村 春彦 (金沢大学)

Recognition of Gaseous Indoor-Air Pollutants Using Multi-Gas Sensory System

Takashi Oyabu, Member (Toyama University of International Studies), Shigeki Hirobayashi, Member, Haruhiko Kimura, Non-member (Kanazawa University)

The housing environment is changing with the change in the energy situation in Japan. The house is gradually becoming an airtight structure to economize energy, and an indoor air-quality is going down. In this study, the gaseous indoor-air pollutants are recognized by a multi-gas sensory system using the OPS5 system which is one of the production systems. The sensory system has six gas sensors. Eight gases are selected as main indoor-air pollutants. They are carbon dioxide, carbon monoxide, ammonia, propane, methane, cigarette smoke, ethanol and formaldehyde gases. The concentration of some of these will be calculated by the system. The sensor outputs are input into a computer every 4 seconds. The recognition processes are done in this interval and a specific kind of pollutant is identified. About twenty rules are employed in this system.

キーワード: gas sensor, multi-gas sensory system, indoor-air pollutants, OPS5, production system

1. ま え が き

社会環境の変遷とともに住環境が大きく変化してきている。特に、熱、空気、音、光などの室内環境の中で空気汚染の悪化が大きな問題となっている⁽¹⁾。また、世帯を構成する家族の減少や老人世帯数の増加による諸問題も生じつつある。特に、老人世帯においては空気・熱環境への適応能力の低下から種々の惨事も発生してきている。本研究においては、人間とりわけ高齢者が認知することが難しいガス状室内空気汚染物の認知について検討した。

製作した認知システムは6種類のセンサからの信号を4秒毎にコンピュータによりモニタし、これらセンサの出力パターンから汚染ガス種を認知するものである⁽²⁾。この認知実現にはプロダクションシステムの一つであるOPS5を用いた⁽³⁾。認知対象ガスは、一酸化炭素(CO)、二酸化炭素(CO₂)、ホルムアルデヒド(HCHO)、アンモニア(NH₃)、メタン(CH₄)、プロパン(C₃H₈)、タバコの煙、エタノール(C₂H₅OH)の8種類である。これらのガスは、家庭内で日常的に発生する可能性の高いものである。また、一酸化炭素、二酸化炭素、プロパン、メタンのように、一般に汚染度の指標に用いられ、発

生すると危険なガスに対してはその濃度も導出する。プロダクションシステムのワーキング・メモリ・エレメント(WME)としては、基本的には6つのセンサからの出力値を用いている。定常状態でのWMEと比較することにより認知処理を遂行する。競合解決のための戦略としてはLEX(Lexicographic-Sort)戦略を採用している。ルールとしては、ガス種認知のためのルールが15個ある。

2. センサと認知システム

実験に供したセンサは全部で6種類ある。熱環境測定のための絶対湿度を検知するためのセンサ(AHS)、一酸化炭素(COS)、二酸化炭素(CO₂S)、可燃性ガス(CGS)、アンモニア(AMS)、窒素酸化物(NXS)をそれぞれ選択的に検知するよう開発されたセンサ(ガス選択性に欠ける点がある)である⁽⁴⁾。この中で、CO₂Sはイオン導電性の高い固体電解質を隔膜として電気化学セルを作成し、センサとして用いたものである。このセルの両端の二酸化炭素の濃度差に応じた電位差が生ずる。CO₂Sは二酸化炭素ガスに対してのみ感度を有し、ガス選択性に優れている。他のガスセンサとAHSは酸化スズ系セラミックガスセンサで、広くガス漏れ警報器等に用いられて




図1 システムの写真

Fig. 1. A photograph of the system.




図2 センサ部の写真

Fig. 2. A photograph of the sensor unit.

いるタイプで信頼性も高い。この他にも温度センサ (TMS) と相対湿度センサ (RHS) が設置してある。このタイプのセンサにおいては、センサ部の母材料が酸化スズであり、それに触媒などを添加することにより各種ガスに対する選択性を賦与する。また、選択性に対してはセンサ構造も大きな要因となる。被検ガスがセンサ近傍に到来するとセンサ素子部の抵抗が減少する。これを素子に直列に接続した負荷抵抗 R_L の両端の電圧 $V (R_L)$ ($0 \sim 5V$) を出力として捕らえ、A-D コンバータを介してコンピュータに入力する。予備実験として、認知対象となる8種類の汚染ガスを200リットルの容器に注入し、各センサの出力パターンを測定した⁽⁵⁾。Fig.1に製作したシステムの写真を示す。また、Fig.2に6個のセンサを組み込んだセンサ部の写真を示す。一般には、二酸化炭素濃度が室内空気汚染度を示す指標として採用され、ビル管理法においては1000ppmを越えないようにという規定値が設けられている。教室等の多くの人が集まる部屋においては、7000ppmを越えることもある。

これらのセンサからの信号をプロダクションシステムの一つであるOPS5により処理する⁽⁶⁾。本来OPS5はワークステーションなどの高い機能を有するコンピュータで稼働させるのが一般的であるが、本研究においては処理で

きるルール数などを減らしたものを用いてパーソナルコンピュータ用に改良したものを用いている。Fig.3にシステムの構成を示す。

3. センサ出力特性

室内環境には、多くのガス状空気汚染物が発生する⁽⁷⁾。その代表的なものがタバコの煙である。危険なガスとしては、燃焼器具の使用などにより発生する一酸化炭素や燃料用ガスであるプロパンなどがある。二酸化炭素は人間の呼吸や燃焼により発生する⁽⁸⁾。アルコール系ガスの代表的なものとしてエタノールを選出している。家庭内においては様々なアルコール系ガスが発生する。すなわち、ヘアースプレーや香水、化粧品などからも発生する。新築住宅においては、新建材などからホルムアルデヒドが、長い時間をかけて室内に放出されることになる。ホルムアルデヒドについては、幼児のアトピーやアレルギー等と関係があることも報じられている⁽⁹⁾。家庭用生ゴミからはアンモニアガスが発生する。メタンガスはそれのみで家庭内で発生することはほとんどない。しかしながら、動植物が水中などで腐敗・発酵して生じ、家庭近くの沼沢などで発生し、室内にも流入する可能性がある。また、マンホール内などでは高濃度になることもあり引火・爆発の危険性がある。これら危険ガスも本システムで検知出来る⁽¹⁰⁾。

これらのガスに対する6つのガスセンサの出力パターンを調べた。Fig.4にタバコの煙に対する各センサの出力特性を示す。これは、測定開始から1分後に火の付いたタバコを実験容器内に置き、5分後に容器を開いてタバコを取り除き、容器内が自然に清浄になるようにしたものである。タバコの煙により、全てのガスセンサの出力が上昇し、容器がオープンになると同時に元のレベルに戻っていく様子がわかる。全てのガスセンサの出力が上昇することから、タバコの煙には様々な種類のガスが含まれていることが伺える。アンモニアガスに対する同様な実験結果をFig.5に示す。これは、200リットルの実験容器に29%のアンモニア水0.05mlを注入し気化させたときの結果である。実験開始1分後に注入し、11分後に容器をオープンし大気レベルに戻したものである。CGS, AMS, NXSが高感度であることがわかる。また、温湿度センサ (TMS, RHS, AHS) とCOS, COS2Sは全く感度を有していない。他の6種類のガスに対しても同様な実験を行った。すなわち、一酸化炭素とプロパンは、実験開始1, 3, 5, 7, 9分後に一定量のガスを注入し、11分後に容器をオープンし各センサの応答特性を調べた。一酸化炭素は10cc (50ppmに相当) ずつを、プロパンは20cc ずつをそれぞれ5回注入した。メタンは50cc ずつ7分までの4回注入し11分後にオープンした。エタノールは99%溶液を1分後に0.15cc 注入し、ホルムアルデヒドは1分後に35%溶液を0.05cc 注入し各々気化させ、メタンと同様に11分後に容器をオープンした。この実験により、各ガスに対するセンサの立上りや立下が




図3 汚染認知システム

Fig. 3. Recognition system of the gaseous indoor-air pollutants.




図4 タバコの煙に対するセンサ出力

Fig. 4. Sensor outputs for cigarette smoke.




図5 アンモニアガスに対するセンサ出力

Fig. 5. Sensor outputs for ammonia gas.

り、大よその濃度特性も把握できる。

これら8つの汚染ガスに対する各ガスセンサの出力パターンをまとめると Table 1 のようになる。この表にお

いては、新築住宅内における各センサの出力パターンも同時にまとめてある。◎は出力が非常に高い、○は高い、△は出力が少しあることを意味する。なお、メタンとプロパ

表1 8種類的气体状室内空気汚染物に対する6つのガスセンサ出力感度。◎, ○, △はそれぞれ感度が非常に高い, 高い, やや感度があることを意味し, ブランクは全く感度がないことを意味している。

Table 1. Output sensitivities of six sensors for eight gaseous indoor-air pollutants. ◎, ○, △ and blank mean very high, high, more or less high and no response, respectively.

	CO	CO ₂	C ₂ H ₅ OH	CH ₄	NH ₃	HCHO	Cigarette Smoke	C ₃ H ₈	New-Built house
CGS	◎		◎	○	◎	◎	◎	◎	○
AMS			◎		◎	○	◎		○
COS	◎		△				◎		
NXS					◎		◎		○
CO2S		◎					◎		
AHS				○					

ンを区別するためにAHSからの出力が必要となる。この表の各ガスセンサ出力のパターンから汚染ガス種を同定することは容易である。すなわち、単に手続型処理により汚染種を認知することが容易であるが、今後、様々な条件下でのガス種の認知を行う必要があるため、拡張性を考慮してOPS5を採用し、ルールによる認知を行った。表からも明らかのように、可燃性ガスセンサ(CGS)は二酸化炭素ガス以外の全ての汚染ガスに感度を有しており、ガス状空気汚染物の検知にはかなり有効であることがわかる。すなわち、まず、CO2SとCGSの出力をモニタしCO2Sのみが上昇すれば、汚染ガスが二酸化炭素であることが判る。しかしながら、住環境においては二酸化炭素ガスが単独で発生することは考えられず、他のガスを伴って発生するのが一般的である。このため、普通はCGSの上昇が汚染ガス発生を認知する引き金となるが、本研究においてはとりあえずTable 1の結果からルールを構成している。新築住宅においては、CGSとAMS, NXSの出力がある。これは、施行会社が建築依頼主により新築し、それを引き渡す時に測定したものである。3軒の測定を行い、感度の差異はあるが、パターンはいずれも同じであった。

4. 汚染ガス種認知のための知識ベース

汚染ガス種を認知するルールは15個ある(付録 start-up 文含む)。また、実験を行った8種類的气体以外は認知しない。ルールの構成は出来る限り基本的なもののみにとどめてある。この知識ベースをベースとして各種のルールを付加することにより、さらに実状に合致する知識ベースを構成し、様々な認知を行えるシステムに発展させることが可能と考えられる⁽¹¹⁾。Fig.4の煙草の煙を認知するルールは、付録の知識ベースのr8である。r8においては、すべてのセンサ出力が上昇した時に” I detected Cigarette Smoke”と表示する。同様に、Fig.5のアンモニアガスを認知するルールはr6である。CGS, AMS, NXSの3つのセンサ出力値が上昇した場合に認知される。6つのセンサからの値は一つのWMEに格納され照合される。汚染ガスの濃度は、一酸化炭素、二酸化炭素、メタン、プロパンに対してのみ出力される。これらは予め行った実験の濃度特性から導出される。ただし、濃度

値はプロダクションシステムの条件照合のコストを押さえるためにドメインで出力することとした。これは、一般家庭においては、各濃度の正確な値よりも、どの程度の影響があり、危険な状態かを示すことが望ましいからである。また、各センサの出力は季節変動により幾分誤差を伴うことによる。

一酸化炭素ガスの濃度に対するCOSの出力特性をFig.6に示す。COガス100ppmまでは、急激に出力が上昇するが、それ以降は飽和する傾向にある。COガスに対しては、20ppmの雰囲気中に8時間、10ppmの雰囲気中に24時間以上さらされないこと、という基準が設けられている。本研究においては、定常状態より0.3V以上出力が上昇した場合には、電子回路により危険レベルにあることを知らせる。

本システムの知識ベースをまとめて付録に示す。また、そのReteネットワークをFig.7に示す。ただし、ネットワーク上の番号は知識ベースの条件要素に付けられたもの




図6 COSの出力のCO濃度依存性

Fig.6. Output of COS as a function of CO concentration.




図7 実験に供した知識ベースのReteネットワーク
Fig.7. Rete network for knowledge base.

と対応する。定常時における各センサの出力値をクラス名 base の WME に取り込み、その値と 4 秒毎に入力してくる各センサの出力値を持つクラス名 value の WME とを比較する。このとき 0.05V 以上センサ出力値が上昇したときに、このセンサが反応したことになる。これは、雑音成分に対する影響を除去するためである。同じ条件下で実験を行えば、ほぼ 100% の認知率が得られている。

5. まとめ

複数のセンサからの出力値をコンピュータに取り込みモニタすることにより、室内における代表的な 8 種類の空気汚染物を認知することを試みた。このとき、汚染ガス種により各ガスセンサの感度が異なることを利用している。また、ガス種を認知したら代表的な 4 種類のガスに対してはおおよその濃度も導出する。認知にはプロダクションシステムの一つである OPS5 を採用した。本研究の範囲内であれば特にプロダクションシステムを用いる必要はないが、実際に利用する場合には様々な環境条件や濃度範囲が考えられるため手続き型処理では不可能となる。このため、知識ベース化することが望ましいと考えられる。

今後は、各種の条件下における検討と汚染ガスの濃度の違いに対するより正確なセンサ特性の把握を行うつもりである。

(平成 9 年 7 月 24 日受付, 同 9 年 9 月 24 日再受付)

文 献

- (1) 瀬沼 勲: "室内環境学", 三共出版 (1990)
- (2) 大藪, 蔡, 石坂, 木村, 松浦, 梶山: "マルチセンサの室内空気汚染物に対する出力特性", Digest of the 20th Chemical Sensor Symposium (電気化学協会・化学センサ研究会) Vol.11

- Supplement A, No.11, pp.49-52 (1995 年 4 月)
- (3) 大藪, 木村: "プロダクションシステムを用いた相対的室内空気汚染の測定", 電学論 E, Vol.117, No.5, pp.243-249 (1997)
- (4) T.Oyabu, C.Kuiqian, S.Ishizaka, Y.Matsuura, H.Kimura: "The Detection of Gaseous Indoor-Air Pollutants Using A Multi-Sensor System", EAST ASIA CONFERENCE ON CHEMICAL SENSORS '95, No.1G04, pp.23-26 (Oct. 5-8, 1995, Xi'an, China)
- (5) 吉浦, 松見, 城戸, 小野寺, 木村, 大藪: "OPS5 によるガス状室内空気汚染物の認知", 平成 7 年度電気関係学会北陸支部連合大会, B-17, p118 (1995 年 9 月)
- (6) 大藪, 木村: "プロダクションシステムによる室内空気汚染度の検知", 電子情報通信学会・安全設計研究会, S95-3, pp.13-18 (1995 年 8 月)
- (7) 大藪, 木村, 石坂: "ガスセンサによる人間のふるまい認知", 電気学会・計測研究会, IM-94-11, pp.77-85 (1994 年 1 月)
- (8) T.Oyabu, H.Kimura: "Indoor Air-Pollution Detector Using Tin-Oxide Gas Sensor", Sensors and Materials, Vol.7, No.6, pp.431-436 (1995)
- (9) 読売新聞, 平成 9 年 4 月 30 日朝刊
- (10) 大藪, 木村: "プロダクションシステムを用いた家庭内災害の同定", 計測自動制御学会論文誌, Vol.30, No.9, pp.1077-1083 (1994)
- (11) 大藪, 木村: "新しい室内空気汚染検知システムの開発", 計測自動制御学会論文誌, Vol.32, No.7, pp.1121-1128(1996)

付 録

;; 追加された各センサの測定値を表すワーキングメモリエレメント

(literalize value ;; クラス名

cgs ;; CGS の値

ams ;; AMS の値

cos ;; COS の値

nxs ;; NXS の値

co2s ;; CO2S の値

ahs) ;; AHS の値

;; 基準となる各センサの測定値を表すワーキングメモリエレメント

(literalize base

cgs

ams

cos

```

    nxs
    co2s
    ahs)
;; CO の電圧差を表すワーキングメモリエレメント
(literalize co
  cos)
;; CO2 の電圧差を表すワーキングメモリエレメント
(literalize co2
  co2s)
;; CH4 の電圧差を表すワーキングメモリエレメント
(literalize ch4
  cgs)
;; プロパンの電圧差を表すワーキングメモリエレメント
(literalize propan
  cgs)
;; CO の対応する濃度と電圧差を表すワーキングメモリエレメント
(literalize cod
  co0 ;;CO の電圧差の下限
  co1 ;;CO の電圧差の上限
  co00 ;;CO の濃度の下限
  co11) ;;CO の濃度の上限
;; CO2 の対応する濃度と電圧差を表すワーキングメモリエレメント
(literalize co2d
  co20 ;;CO2 の電圧差の下限
  co21 ;;CO2 の電圧差の上限
  co200 ;;CO2 の濃度の下限
  co211) ;;CO2 の濃度の上限
;; CH4 の対応する濃度と電圧差を表すワーキングメモリエレメント
(literalize ch4d
  cgs0 ;;CGS の電圧差の下限
  cgs1 ;;CGS の電圧差の上限
  cgs00 ;;CH4 の濃度の下限
  cgs11) ;;CH4 の濃度の上限
;; プロパンの対応する濃度と電圧差を表すワーキングメモリエレメント
(literalize propan
  cgs0 ;;CGS の電圧差の下限
  cgs1 ;;CGS の電圧差の上限
  cgs000 ;;プロパンの濃度の下限
  cgs111) ;;プロパンの濃度の上限

(p startup
  (start)
-->
  (make base ^cgs (compute(accept)+0.05)
    ^ams (compute(accept)+0.05)
    ^cos (compute(accept)+0.05)
    ^nxs (compute(accept)+0.05)
    ^co2s (compute(accept)+0.05)
    ^ahs (compute(accept)+0.05)))
  (make cod ^co0 0 ^co1 2.02
    ^co00 0 ^co11 50)
  (make cod ^co0 2.02 ^co1 2.27
    ^co00 50 ^co11 100)
  (make cod ^co0 2.27 ^co1 2.42
    ^co00 100 ^co11 150)
  (make cod ^co0 2.42 ^co1 2.49
    ^co00 150 ^co11 200)
  (make cod ^co0 2.49 ^co1 2.53
    ^co00 200 ^co11 250)
  (make cod ^co0 0 ^co1 2.02
    ^co00 0 ^co11 50)
  (make co2d ^co20 0 ^co21 0.11
    ^co200 300 ^co211 550)
  (make co2d ^co20 0.11 ^co21 0.42
    ^co200 550 ^co211 800)
  (make co2d ^co20 0.42 ^co21 0.84
    ^co200 800 ^co211 1050)
  (make co2d ^co20 0.84 ^co21 1.25
    ^co200 1050 ^co211 1300)
  (make co2d ^co20 1.25 ^co21 1.46
    ^co200 1300 ^co211 1550)
  (make co2d ^co20 1.46 ^co21 1.63
    ^co200 1550 ^co211 1800)
  (make ch4d ^cgs0 0 ^cgs1 0.4
    ^cgs00 0 ^cgs11 250)
  (make ch4d ^cgs0 0.4 ^cgs1 0.6
    ^cgs00 250 ^cgs11 500)
  (make ch4d ^cgs0 0.6 ^cgs1 0.74
    ^cgs00 500 ^cgs11 750)
  (make ch4d ^cgs0 0.74 ^cgs1 0.88
    ^cgs00 750 ^cgs11 1000)
  (make propan ^cgs0 0 ^cgs1 0.576
    ^cgs000 0 ^cgs111 100)
  (make propan ^cgs0 0.576 ^cgs1 0.8
    ^cgs000 100 ^cgs111 200)
  (make propan ^cgs0 0.8 ^cgs1 0.959
    ^cgs000 200 ^cgs111 300)
  (make propan ^cgs0 0.959 ^cgs1 1.079
    ^cgs000 300 ^cgs111 400)
  (make propan ^cgs0 1.079 ^cgs1 1.173
    ^cgs000 400 ^cgs111 500)

(p r1
  (base ^cgs <a> ^ams <b> ^cos <c>
    ^nxs <d> ^co2s <e> ^ahs <h>))
-->
  (make value ^cgs (accept) ^ams (accept)
    ^cos (accept) ^nxs (accept)
    ^co2s (accept) ^ahs (accept)))

(p r2
  (base ^cgs <a> ^ams <b> ^cos <c>
    ^nxs <d> ^co2s <e> ^ahs <h>))
  (value ^cgs > <a> ^ams <= <b> ^cos {<f> > <c>}
    ^nxs <= <d> ^co2s <= <e> ^ahs <= <h>))
-->
  (write (crlf)"I detected CO.")
  (make co ^cos (compute <f> - <c> + 0.05))
  (remove 2))

(p r21
  (co ^cos <a>)
  (cod ^co0 < <a> ^co1 >= <a> ^co00 <b> ^co11 <c>))
-->
  (write <b> ~ <c> (crlf))
  (remove 1)
  (make value ^cgs (accept) ^ams (accept)
    ^cos (accept) ^nxs (accept)
    ^co2s (accept) ^ahs (accept)))

(p r3
  (base ^cgs <a> ^ams <b> ^cos <c>
    ^nxs <d> ^co2s <e> ^ahs <h>))
  (value ^cgs <= <a> ^ams <= <b> ^cos <= <c>
    ^nxs <= <d> ^co2s {<f> > <e>}
    ^ahs <= <h>))
-->
  (write (crlf)"I detected CO2."))

```

```

(make co2 ^co2s (compute <f> - <e> +0.05))
(remove 2))

(p r31
  (co2 ^co2s <a>)
  (co2d ^co20 < <a> ^co21 >= <a>
    ^co200 <b> ^co211 <c>))
-->
(write <b> ~ <c> (crlf))
(remove 1)
(make value ^cgs (accept) ^ams (accept)
  ^cos (accept) ^nxs (accept)
  ^co2s (accept) ^ahs (accept)))

(p r4
  (base ^cgs <a> ^ams <b> ^cos <c>
    ^nxs <d> ^co2s <e> ^ahs <h>))
(value ^cgs > <a> ^ams > <b> ^cos > <c>
  ^nxs <= <d> ^co2s <= <e> ^ahs <= <h>))
-->
(write (crlf)"I detected C2H5OH.")
(remove 2)
(make value ^cgs (accept) ^ams (accept)
  ^cos (accept) ^nxs (accept)
  ^co2s (accept) ^ahs (accept)))

(p r5
  (base ^cgs <a> ^ams <b> ^cos <c>
    ^nxs <d> ^co2s <e> ^ahs <h>))
(value ^cgs {<f> > <a>} ^ams <= <b> ^cos <= <c>
  ^nxs <= <d> ^co2s <= <e> ^ahs > <h>))
-->
(write (crlf)"I detected CH4.")
(make ch4 ^cgs (compute <f> - <a> + 0.05))
(remove 2))

(p r51
  (ch4 ^cgs <a>)
  (ch4d ^cgs0 < <a> ^cgs1 >= <a>
    ^cgs00 <b> ^cgs11 <c>))
-->
(write <b> ~ <c> (crlf))
(remove 1)
(make value ^cgs (accept) ^ams (accept)
  ^cos (accept) ^nxs (accept)
  ^co2s (accept) ^ahs (accept)))

(p r6
  (base ^cgs <a> ^ams <b> ^cos <c>
    ^nxs <d> ^co2s <e> ^ahs <h>))
(value ^cgs > <a> ^ams > <b> ^cos <= <c>
  ^nxs > <d> ^co2s <= <e> ^ahs <= <h>))
-->
(write (crlf)"I detected NH3.")
(remove 2)
(make value ^cgs (accept) ^ams (accept)
  ^cos (accept) ^nxs (accept)
  ^co2s (accept) ^ahs (accept)))

(p r7
  (base ^cgs <a> ^ams <b> ^cos <c>
    ^nxs <d> ^co2s <e> ^ahs <h>))
(value ^cgs > <a> ^ams > <b> ^cos <= <c>
  ^nxs <= <d> ^co2s <= <e> ^ahs <= <h>))
-->
(write (crlf)"I detected HCHO.")

```

```

(remove 2)
(make value ^cgs (accept) ^ams (accept)
  ^cos (accept) ^nxs (accept)
  ^co2s (accept) ^ahs (accept)))

(p r8
  (base ^cgs <a> ^ams <b> ^cos <c>
    ^nxs <d> ^co2s <e> ^ahs <h>))
(value ^cgs > <a> ^ams > <b>
  ^cos > <c> ^nxs > <d> ^co2s > <e>
  ^ahs <= <h>))
-->
(write (crlf)"I detected Cigarette Smoke.")
(remove 2)
(make value ^cgs (accept) ^ams (accept)
  ^cos (accept) ^nxs (accept)
  ^co2s (accept) ^ahs (accept)))

(p r9
  (base ^cgs <a> ^ams <b> ^cos <c>
    ^nxs <d> ^co2s <e> ^ahs <h>))
(value ^cgs {<f> > <a>} ^ams <= <b> ^cos <= <c>
  ^nxs <= <d> ^co2s <= <e> ^ahs <= <h>))
-->
(write (crlf)"I detected Propane.")
(make propan ^cgs (compute <f> - <a> + 0.05))
(remove 2))

(p r91
  (propan ^cgs <a>)
  (propan ^cgs0 < <a> ^cgs1 >= <a>
    ^cgs00 <b> ^cgs11 <c>))
-->
(write <b> ~ <c> (crlf))
(remove 1)
(make value ^cgs (accept) ^ams (accept)
  ^cos (accept) ^nxs (accept)
  ^co2s (accept) ^ahs (accept)))

(p r10
  (value ^cgs <a> ^ams <b> ^cos <c>
    ^nxs <d> ^co2s <e> ^ahs <h>))
-->
(remove 1)
(make value ^cgs (accept) ^ams (accept)
  ^cos (accept) ^nxs (accept)
  ^co2s (accept) ^ahs (accept)))

```

大 藪 多可志 (正員) 1973 年, 工学院大学工学研究科修士課程修了。75 年早稲田大学第二文学部英文学科卒業。73 年電気音響 (株) 技術研究所勤務。80 年金沢女子短期大学講師, 助教授, 教授を経て, 91 年富山国際大学人文学部社会学科助教授, 94 年教授。ガスセンサシステムに関する研究に従事 (工学博士)。



広 林 茂 樹 (正員) 94年3月工学院大学工学研究科修士課程修



了。95年3月同大学博士課程中退。95年4月金沢大学工学部助手。主として、室内音響、音響波動・振動論、音響信号処理、音場制御、流体伝達系のモデル化に関する研究に従事。電子情報通信学会、計測自動制御学会、日本音響学会、米国音響学会各会員

木 村 春 彦 (非会員) 1974年、東京電機大学工学部応用理化学



学科卒業。79年東北大学工学研究科博士(情報処理)課程修了。同年富士通(株)勤務。80年金沢女子短期大学講師。84年金沢大学経済学部助教授を経て、現在、同大学工学部電気・情報工学科教授。その間、最適コード変換、プロダクションシステムの高速化の研究に従事(工学博士)。